

ECCE 550/ME550 Simulation Lab 3: ANSYS Simulation of a Bistable Mechanism

1 Introduction

Finite-element analysis allows us to model compliant structures with greater accuracy than is possible with pseudo-rigid-body models. In this lab you will use ANSYS to do a finite-element analysis of the bistable mechanism we have been studying.

2 Finite-Element Analysis

Attached is an ANSYS batch file that simulates a cantilever beam with two cross sectional areas. A long rigid segment is attached to ground via a small thin flexible segment. The beam's response is modeled as a displacement is applied to the end. The displacement of the end and the corresponding reaction force is output to a text file for latter plotting in MATLAB. A graphic displaying the deformed shape is finally plotted in the ANSYS GUI.

Use this batch file as a guide to write your own batch file that simulates a fully compliant bistable mechanism. As you follow the batch file explore each of the commands that are issue by using the ANSYS help files. Your batch file should output a text file that lists corresponding displacements and forces, and should plot the deformed shape.

Compare the finite-element solution to your MATLAB solution for the input parameters listed below for a SU-8 device.

3 Deliverables

Turn in a professional memo that includes a screenshot of the deformed shape along with a plot comparing the finite-element solution with your MATLAB solution. Discuss why they might be different.

Table 1: FCBM Parameters.

<i>Parameter</i>	<i>SU-8</i>	
w	7	μm
$w_{\text{centersection}}$	35	μm
t	30	μm
ℓ	100	μm
r_{10}	298	μm
θ_0	0.105	radians
E	4.4	GPa

4 Example Batch File

```

!*****
! Created: 3 Oct 2006
! By: Rob Messenger
!
! This ANSYS batch fileuses BEAM3 elements to analyse
! a stepped cantilever beam.
!
! /|          |
! /|          \|/
! /|=====-----
! /|  ll,hh  1,h
! /|
!
!*****
FINISH
/CLEAR,START
/TITLE, Stepped Cantilever Beam
/FILENAME, cantilever, 1
/Prep7
!*****
! Input Parameters
!*****

ll = 10
hh = 1
l = 75
h = 100

```

```

t = 2
Ey = 165
Pr = 0.33

SegmentsBig = 10*(ll)
*IF,SegmentsBig,LT,100,THEN
  SegmentsBig = 100
*ENDIF

SegmentsSmall = 10*(l)
*IF,SegmentsSmall,LT,100,THEN
  SegmentsSmall = 100
*ENDIF

TShearConst = 6/5
Nonlinear = 1
Loadsteps = 20
dY = 0.2*(ll+l)

!*****
! Model Setup
!*****
ii = t*hh*hh*hh/12
i = t*h*h*h/12

K,1,0,0,0
K,2,ll,0,0
K,3,ll+l,0,0

L,1,2

L,2,3

LESIZE,1,,SegmentsBig
LESIZE,2,,SegmentsSmall

!*** Element Type: BEAM3 (2D Elastic)
ET,1,BEAM3
R,1,hh*t,ii,hh,TShearConst,0,0
R,2,h*t,i,h,TShearConst,0,0
MP,EX,1,Ey

```

```

MP,PRXY,1,Pr

!*** Mesh
!lsel,s,line,,1,,1
type,1
real,1
mat,1
lmesh,1

allsel,all
lsel,s,line,,2,,1
type,1
real,2
mat,1
lmesh,2

allsel,all

! *** Get Node Number at Keypoint 3
KSEL,s,kp,,3
NSLK,s
*GET,nkp3,node,0,num,max
allsel,all

FINISH

!*****
! Loads
!*****
/SOLU

*IF,Nonlinear,EQ,1,THEN
  NLGEOM,1
*ENDIF
ANTYPE,0

!*** Boundry Conditions
DK,1,all

!*** Define Loads
*Do,mm,1,Loadsteps,1

```

```

DK,3,UY,mm*dY/Loadsteps
lswrite,mm
*ENDDO

lssolve,1,Loadsteps
FINISH

!*****
! Output
!*****
/POST1
*DIM,Ydis,TABLE,Loadsteps
*DIM,Yforce,TABLE,Loadsteps

*DO,mm,1,Loadsteps,1
SET,mm

*GET,disY,NODE,nkp3,U,Y
*SET,Ydis(mm),disY

*GET,forceY,NODE,nkp3,RF,FY
*SET,Yforce(mm),forceY
*ENDDO

/output,dispVforce.txt
*VWRITE,Ydis(1),Yforce(1)
%16.8G %16.8G
/output

SET,LAST
PLDISP,0
/AUTO,1
/REP

FINISH
/eof

```